Trigonometric Polynomial Approximations

Continuous

```
> restart;
```

We find the **continuous least squares trigonometric polynomial** S_{10}(x) for f(x)=e^x on [-Pi, Pi].

We first enter our value for n and our function.

```
> n:=10;
```

$$n := 10$$

```
> f:=x->exp(x);
```

$$f := x \rightarrow e^x$$

We comput the Fourier coefficients a_k and b_k.

```
> for k from 0 to n do
  a[k]:=1/Pi*Int(f(x)*cos(k*x),x=-Pi..Pi);
  a[k]:=evalf(a[k]);
  od;
```

$$a_0 := \frac{\int_{-\pi}^{\pi} e^x \, dx}{\pi}$$

$$a_0 := 7.352155817$$

$$a_1 := \frac{\int_{-\pi}^{\pi} e^x \cos(x) \, dx}{\pi}$$

$$a_1 := -3.676077910$$

$$a_2 := \frac{\int_{-\pi}^{\pi} e^x \cos(2\,x) \, dx}{\pi}$$

$$a_2 := 1.470431164$$

$$a_3 := \frac{\int_{-\pi}^{\pi} e^x \cos(3\,x) \, dx}{\pi}$$

$$a_3 := -0.7352155817$$

$$a_4 := \frac{\int_{-\pi}^{\pi} e^x \cos(4\,x) \, dx}{\pi}$$

$$a_4 := 0.4324797542$$

$$a_5 := \frac{\int_{-\pi}^{\pi} e^x \cos(5\,x)\,dx}{\pi}$$

$$a_5 := -0.2827752238$$

$$a_6 := \frac{\int_{-\pi}^{\pi} e^x \cos(6\,x)\,dx}{\pi}$$

$$a_6 := 0.1987069140$$

$$a_7 := \frac{\int_{-\pi}^{\pi} e^x \cos(7\,x)\,dx}{\pi}$$

$$a_7 := -0.1470431164$$

$$a_8 := \frac{\int_{-\pi}^{\pi} e^x \cos(8\,x)\,dx}{\pi}$$

$$a_8 := 0.1131100895$$

$$a_9 := \frac{\int_{-\pi}^{\pi} e^x \cos(9\,x)\,dx}{\pi}$$

$$a_9 := -0.08966043682$$

$$a_{10} := \frac{\int_{-\pi}^{\pi} e^x \cos(10\,x)\,dx}{\pi}$$

$$a_{10} := 0.07279362198$$

```
> for k from 1 to n-1 do
  b[k]:=1/Pi*Int(f(x)*sin(k*x),x=-Pi..Pi);
  b[k]:=evalf(b[k]);
  od;
```

$$b_1 := \frac{\int_{-\pi}^{\pi} e^x \sin(x)\,dx}{\pi}$$

$$b_1 := 3.676077910$$

$$b_2 := \frac{\int_{-\pi}^{\pi} e^x \sin(2\,x)\,dx}{\pi}$$

$$b_2 := -2.940862328$$

$$b_3 := \frac{\int_{-\pi}^{\pi} e^x \sin(3\,x)\,dx}{\pi}$$

$$b_3 := 2.205646746$$

$$b_4 := \frac{\int_{-\pi}^{\pi} e^x \sin(4\,x)\,dx}{\pi}$$

$$b_4 := -1.729919016$$

$$b_5 := \frac{\int_{-\pi}^{\pi} e^x \sin(5\,x)\,dx}{\pi}$$

$$b_5 := 1.413876119$$

$$b_6 := \frac{\int_{-\pi}^{\pi} e^x \sin(6\,x)\,dx}{\pi}$$

$$b_6 := -1.192241484$$

$$b_7 := \frac{\int_{-\pi}^{\pi} e^x \sin(7\,x)\,dx}{\pi}$$

$$b_7 := 1.029301815$$

$$b_8 := \frac{\int_{-\pi}^{\pi} e^x \sin(8\,x)\,dx}{\pi}$$

$$b_8 := -0.9048807162$$

$$b_9 := \frac{\int_{-\pi}^{\pi} e^x \sin(9\,x)\,dx}{\pi}$$

$$b_9 := 0.8069439313$$

We now form the degree *n* trigonometric polynomial over the interval [-Pi, Pi] in the variable x. We name this polynomial S_n. We first reset two variables.

```
> n:='n';k:='k';
```

$$n := n$$

$$k := k$$

```
> S[n]:='a[0]'/2+a[n]*cos(n*x)+Sum(a[k]*cos(k*x)+b[k]*sin(k*x),k=1..n-1);
```

$$S_n := \frac{1}{2}\, a_0 + a_n \cos(n\,x) + \sum_{k=1}^{n-1} \left(a_k \cos(k\,x) + b_k \sin(k\,x) \right)$$

We now look specifically at S_{10}.

```
> S[10]:=unapply(a[0]/2+a[10]*cos(10*x)+add(a[k]*cos(k*x)+b[k]*sin(k*x),k=1..10-1),x);
```

$$S_{10} := x \to 3.676077908 + 0.07279362198 \cos(10\,x) - 3.676077910 \cos(x) + 3.676077910 \sin(x) + 1.470431164 \cos(2\,x) - 2.940862328 \sin(2\,x) - 0.7352155817 \cos(3\,x) + 2.205646746 \sin(3\,x) + 0.4324797542 \cos(4\,x) - 1.729919016 \sin(4\,x) - 0.2827752238 \cos(5\,x) + 1.413876119 \sin(5\,x) + 0.1987069140 \cos(6\,x) - 1.192241484 \sin(6\,x) - 0.1470431164 \cos(7\,x) + 1.029301815 \sin(7\,x) + 0.1131100895 \cos(8\,x) - 0.9048807162 \sin(8\,x) - 0.08966043682 \cos(9\,x) + 0.8069439313 \sin(9\,x)$$

We graph the function (red) and the trigonometric polynomial (green) on the interval [-Pi, Pi].

```
> plot({f(x),S[10](x)},x=-Pi..Pi);
```



Notice the less than perfect fit. Increasing the degree of the polynomial will improve the fit, but a real good fit may necessitate a very high degree. This should not be surprising since we are fitting a non-period function with a sum of periodic functions.

Discrete

We do Example 2 on page 373.

We first enter the value of m such that we have 2m data points (x_j, y_j). We also enter the degree n of our trigonometric polynomial with n<m. The greater we make n the more accurate our polynomial should be.

```
> m:=5;n:=3;
```

$$m := 5$$

$$n := 3$$

We input the endpoints of our closed interval [A,B].

```
> A:=0;B:=2;
```

$$A := 0$$

$$B := 2$$

We form the array x of equally spaced x values.

```
> for j from 0 to 2*m-1 do
  x[j]:=A+j*(B-A)/(2*m)
```

```
od;
```

$$x_0 := 0$$

$$x_1 := \frac{1}{5}$$

$$x_2 := \frac{2}{5}$$

$$x_3 := \frac{3}{5}$$

$$x_4 := \frac{4}{5}$$

$$x_5 := 1$$

$$x_6 := \frac{6}{5}$$

$$x_7 := \frac{7}{5}$$

$$x_8 := \frac{8}{5}$$

$$x_9 := \frac{9}{5}$$

We enter our y values into the y array either by using a function that generates them or by putting them into a list L and using a for loop to place them into the array.

```
> f:=x->x^4-3*x^3+2*x^2-tan(x*(x-2));
> for j from 0 to 2*m-1 do
  y[j]:=evalf(f(x[j]))
  od;
```

$$f := x \rightarrow x^4 - 3\,x^3 + 2\,x^2 - \tan(x\,(x-2))$$

$$y_0 := 0.$$

$$y_1 := 0.4340028516$$

$$y_2 := 0.8981438222$$

$$y_3 := 1.317232349$$

$$y_4 := 1.581957491$$

$$y_5 := 1.557407725$$

$$y_6 := 1.197957491$$

$$y_7 := 0.6452323490$$

$$y_8 := 0.1301438222$$

$$y_9 := -0.1419971484$$

We use a linear transformation z = T(x) = -Pi+(2Pi)*(x-A)/(B-A) to transform the interval [A, B] into the interval [-Pi, Pi]. We use the linear transformation to transform each x_j in [A, B] into the corresponding

z_j in [-Pi, Pi].

```
> eq:=z=-Pi+(2*Pi)*(x-A)/(B-A);
```

$$eq := z = -\pi + \pi\, x$$

```
> for j from 0 to 2*m-1 do
  z[j]:=-Pi+(2*Pi)*(x[j]-A)/(B-A)
  od;
```

$$z_0 := -\pi$$

$$z_1 := -\frac{4}{5}\pi$$

$$z_2 := -\frac{3}{5}\pi$$

$$z_3 := -\frac{2}{5}\pi$$

$$z_4 := -\frac{1}{5}\pi$$

$$z_5 := 0$$

$$z_6 := \frac{1}{5}\pi$$

$$z_7 := \frac{2}{5}\pi$$

$$z_8 := \frac{3}{5}\pi$$

$$z_9 := \frac{4}{5}\pi$$

We next find the coefficients a_k and b_k.

```
> for k from 0 to n do
  a[k]:=evalf(1/m*add(y[j]*cos(k*z[j]),j=0..2*m-1));
  od;
```

$$a_0 := 1.524016151$$

$$a_1 := 0.7717690394$$

$$a_2 := 0.0174227900$$

$$a_3 := 0.0065672733$$

```
> for k from 1 to n-1 do
  b[k]:=evalf(1/m*add(y[j]*sin(k*z[j]),j=1..2*m-1));
  od;
```

$$b_1 := -0.3867590453$$

$$b_2 := 0.04780604708$$

We now form the trigonometric polynomial over the interval [-Pi, Pi] in the variable z. We name this polynomial s.

```
> s:=z->a[0]/2+a[n]*cos(n*z)+sum('a[k]*cos(k*z)+b[k]*sin(k*z)','k'=
  1..n-1);
```

```
> s:=unapply(s(z),z);;
```

$$s := z \to \frac{1}{2}\, a_0 + a_n \cos(n\, z) + \sum_{'k' = 1}^{n-1} 'a_k \cos(k\, z) + b_k \sin(k\, z)'$$

$$s := z \to 0.7620080755 + 0.0065672733 \cos(3\, z) + 0.7717690394 \cos(z) - 0.3867590453 \sin(z) + 0.0174227900 \cos(2\, z) + 0.04780604708 \sin(2\, z)$$

We translate the trigonometric polynomial back to our original interval [A, B].

```
> Z:=rhs(eq);
> S:=unapply(s(Z),x);
```

$$Z := -\pi + \pi\, x$$

$$S := x \to 0.7620080755 - 0.0065672733 \cos(3\, \pi\, x) - 0.7717690394 \cos(\pi\, x) + 0.3867590453 \sin(\pi\, x) + 0.0174227900 \cos(2\, \pi\, x) + 0.04780604708 \sin(2\, \pi\, x)$$

We compute the least squares error.

```
> lserror:=evalf(add(abs(y[j]-S(x[j]))^2,j=0..2*m-1));
```

$$lserror := 0.0009841079475$$

We graph the function (red) and the trigonometric polynomial (green) on the interval [A, B].

```
> plot({f(x),S(x)},x=A..B);
```